# The Â class

Let $V \to X$ be a real, oriented vector bundle of rank $4\,n$, let $p_i$ be its Pontryagin classes, and let $p = 1 + p_1 + p_2 + \cdots + p_{2\,n}$ be the total Pontryagin class. We introduce $2\,n$ formal variables $y_i$ so that the Pontryagin classes are the elementary symmetric polynomials in the $y_i$. Then we can write the Pontryagin and Euler classes formally as:

$$p\,(V) = \prod (y_i + 1)$$

$$e\,(V) = \prod \sqrt{y_i}\,.$$

If $V$ admits a spin structure, and $S^+(V)$, $S^-(V)$ are the spin bundles, then we write $s(V)$ for the characteristic class

$$s\,(V) = \mathrm{ch}\,(S^+\,(V) - S^-\,(V))\,.$$

This we can also express in terms of the roots $y_i$:

$$s\,(V) = \prod 2\,\mathrm{Sinh}\,(\sqrt{y_i}\,/\,2)\,.$$

Both $s(V)$ and $e(V)$ are multiplicative under Whitney sum. Note also that we can expand the class $s(V)$ formally in terms of the Euler class and Pontryagin classes. So a spin structure is not needed.

Now let $X$ be a smooth, oriented manfiold, and $T$ is its tangent bundle. The ratio of the formal expressions for $e(T)$ and $s(T)$ is the $\hat{A}$-class:

$$\hat{A}\,(X) = \prod \frac{\sqrt{y_i}\,/\,2}{\mathrm{Sinh}\,(\sqrt{y_i}\,/\,2)}\,.$$

This is to be interpreted as a symmetric formal power series in the $y_i$, which is then to be expressed in terms of the Pontryagin classes (the elementary symmetric polynomials). Note that the square roots disappear in the expansion, so we *do* get a power series in the $y_i$. In particular, since we end up with an expression in the Pontryagin classes, an orientation is not needed in the end.

In this notebook, we will compute the formal power series and express it in terms of the Pontryagin classes. Then we will calculate the $\hat{A}$-genus (the evaluation of the $\hat{A}$-class on the fundamental class of the manifold) for hypersurfaces in complex projective spaces.

```
<< Algebra`SymmetricPolynomials`
```

The power series for the $\hat{A}$-class is the product of terms like these:

```
GenFun[y_] = (√y / 2) / Sinh[√y / 2];
```

We ask *Mathematica* to compute the power series as far as the $n$th power. We put $n = 6$. We convert the truncated power series to a normal polynomial expression using *Mathematica's* function **Normal**:

```
n = 6;
a[y_] = Series[GenFun[y], {y, 0, n}] // Normal
```

$$1 - \frac{y}{24} + \frac{7\,y^2}{5760} - \frac{31\,y^3}{967680} + \frac{127\,y^4}{154828800} - \frac{73\,y^5}{3503554560} + \frac{1414477\,y^6}{2678117105664000}$$

Here is the power series for the $\hat{A}$-class. We put in a dummy variable $t$ to keep track of the total degree, and we throw out terms past $t^n$:

```
aroofLong = ∏_{i=1}^{n} a[t y_i];
```

```
aroof = Series[aroofLong, {t, 0, n}] // Normal;
```

We express the $\hat{A}$-class in terms of the Pontryagin classes. The *Mathematica* function **SymmetricReduction** is used to express a symmetric polynomial in terms of the elementary symmetric polynomials.

```
yVec = Table[y_i, {i, 1, n}];
pVec = Table[p_i, {i, 1, n}];
Â = SymmetricReduction[aroof, yVec, pVec][[1]];
```

We exract the terms of the $\hat{A}$-class in each dimension. We put each expressionn over a common denominator, and then print them out.

```
Â_i_ := ExpandNumerator[Simplify[Coefficient[Â, t, i]]]
For[i = 1, i < n + 1, i++, Print[Subscript[OverHat["A"], i], "=", Â_i]];
```

$\hat{A}_1 = -\frac{p_1}{24}$

$\hat{A}_2 = \frac{7\,p_1^2 - 4\,p_2}{5760}$

$\hat{A}_3 = \frac{-31\,p_1^3 + 44\,p_1\,p_2 - 16\,p_3}{967680}$

$\hat{A}_4 = \frac{381\,p_1^4 - 904\,p_1^2\,p_2 + 208\,p_2^2 + 512\,p_1\,p_3 - 192\,p_4}{464486400}$

$\hat{A}_5 = \frac{-2555\,p_1^5 + 8584\,p_1^3\,p_2 - 4976\,p_1\,p_2^2 - 5856\,p_1^2\,p_3 + 2688\,p_2\,p_3 + 3392\,p_1\,p_4 - 1280\,p_5}{122624409600}$

$\hat{A}_6 = \frac{1}{2678117105664000}$
$(1414477\,p_1^6 - 6161812\,p_1^4\,p_2 + 6099760\,p_1^2\,p_2^2 - 769728\,p_2^3 + 4636288\,p_1^3\,p_3 - 5362176\,p_1\,p_2\,p_3 + 719872\,p_3^2 - 3217728\,p_1^2\,p_4 + 1476352\,p_2\,p_4 + 1872384\,p_1\,p_5 - 707584\,p_6)$

We look at hypersurfaces of degree $d$ in complex projective $k$-space. Here is the total Chern class of projective space and of the hypersurface, in terms of the two-dimensional generator $h$.

```
cp[k_, h_] = (1 + h)^(k+1);
```

```
hyp[d_, k_, h_] := Series[(1 + h)^(k+1)/(1 + d h), {h, 0, k - 1}] // Normal
```

For example, here is the total Chern class of the *K*3 surface, a quartic in 3-space:

```
hyp[4, 3, h]
```

```
1 + 6 h^2
```

Here is the total Pontryagin class of the hypersurface, computed from the total Chern class. The "long" version has meaningless terms beyond the dimension of the manifold.

```
longPontHyp[d_, k_, g_] := Expand[
  (Evaluate[ hyp[d, k, h ]] /. h → I * g) * (Evaluate[ hyp[d, k, h ]] /. h → -I * g)]
pontHyp[d_, k_, g_] := Series[longPontHyp[d, k, g], {g, 0, k - 1}] // Normal
```

Here are the individual Pontryagin classes; we extract only the integer coefficient.

```
P_i_[d_, k_] := Coefficient[pontHyp[d, k, g],  g^(2 i)]
```

Here is the $\hat{A}$-genus of the hypersurface. The factor of $d$ comes from evaluating $h^{k-1}$ on the fundamental class of the hypersurface.

```
AroofHyp[d_, k_] := d Â_(k-1)/2 /.
  Table[ p_i → P_i[d, k], {i, 1, n}]
```

With $d = k + 1$ (the Calabi-Yau case, meaning that $c_1$ is zero), we get $\hat{A}$-genus 2. One way to explain this is to say that the $\hat{A}$-genus is the index of the Dirac operator, which coincides with the index of the $\overline{\partial}$-complex, acting on $(0, q)$-forms in the Calabi-Yau case. So the $\hat{A}$-genus is also the holomorphic Euler characteristic:

$$\chi = \sum \dim H^{0,q}(X).$$

On the other hand, since $X$ is a hypersurface, the Lefschetz theorem tells us that $\dim H^{0,q}$ is zero except for $q = 0$, and $q = k$. For these values, the dimension is 1 (in the latter case, because $K_X$ is trivial). We check that we get $\hat{A}$-genus 2 for a Calabi-Yau hypersurface in 7-space:

```
AroofHyp[8, 7]
```

```
2
```

If the manifold is spin, the $\hat{A}$-genus is an integer. In our case, $k$ needs to be odd in order for the real dimension of the hypersurface to be a multiple of 4. The hypersurface is then spin if and only if $d$ is even. If the manifold is not spin, the $\hat{A}$-genus is not necessarily an integer. Its failure is only in the power of two, however. We check that we get integers for hypersurfaces of even degree in 5-space:

```
Table[AroofHyp[2 e, 5], {e, 1, 10}]
```

```
{0, 0, 2, 12, 42, 112, 252, 504, 924, 1584}
```

We observe that for hypersurfaces of odd degree, then denominator is a power of two:

```
Table[AroofHyp[2 e - 1, 5], {e, 1, 10}]
```

$$\left\{\frac{3}{128}, -\frac{7}{128}, \frac{63}{128}, \frac{693}{128}, \frac{3003}{128}, \frac{9009}{128}, \frac{21879}{128}, \frac{46189}{128}, \frac{88179}{128}, \frac{156009}{128}\right\}$$

Another phenomenon we can see in the even case is that, for hypersurfaces of even degree, smaller than the degree of the Calabi-Yau hypersurface, the $\hat{A}$-genus is zero. An explanation is that these manifolds have metrics of positive scalar curvature, and this ensures that the index of the Dirac operator is zero. Note that, in the same range, the $\hat{A}$-genus of the hypersurfaces of odd degree can be non-zero. Here is the same phenomenon exhibited for hypersurfaces in 7-space:

```
Table[AroofHyp[d, 7], {d, 1, 8}]
```

$$\left\{-\frac{5}{1024}, 0, \frac{9}{1024}, 0, -\frac{33}{1024}, 0, \frac{429}{1024}, 2\right\}$$

For $k = 2, 4$ and 6, we give the $\hat{A}$-genus as a function of the degree $d$ of the hypersurface:

```
Expand[AroofHyp[d, 3]]
```

$$-\frac{d}{6} + \frac{d^3}{24}$$

```
Expand[AroofHyp[d, 5]]
```

$$\frac{d}{30} - \frac{d^3}{96} + \frac{d^5}{1920}$$

```
Expand[AroofHyp[d, 7]]
```

$$-\frac{d}{140}+\frac{7\,d^3}{2880}-\frac{d^5}{5760}+\frac{d^7}{322560}$$

With hindsight, we can cut straight to the answers without going through the calculation. We can verify that, for hypersurfaces of degree $d$ in $(k+1)$-space, the answer must be an odd polynomial in $d$, of degree $k+1$. Furthermore, we know the value of the $\hat{A}$-genus for all even values $d \leq k+1$ (either 0 or 2). So we know enough values to determine the polynomial. We verify this for $k=6$, by finding the polynomial whose value is 2 at $d=8$ and whose value is 0 for even integers in between $-6$ and 6:

$$\texttt{Expand}\left[2 * \texttt{Product}\left[\frac{(d-e)}{(8-e)},\ \{e, -6, 6, 2\}\right]\right]$$

$$-\frac{d}{140}+\frac{7\,d^3}{2880}-\frac{d^5}{5760}+\frac{d^7}{322560}$$

This is indeed the same polynomial we obtained above.